

# NAG Fortran Library Routine Document

## F08YUF (ZTGSEN)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F08YUF (ZTGSEN) reorders the generalized Schur factorization of a complex matrix pair in generalized Schur form, so that a selected cluster of eigenvalues appears in the leading elements on the diagonal of the generalized Schur form. The routine also, optionally, computes the reciprocal condition numbers of the cluster of eigenvalues and/or corresponding deflating subspaces.

### 2 Specification

```

SUBROUTINE F08YUF ( IJOB, WANTQ, WANTZ, SELECT, N, A, LDA, B, LDB, ALPHA,
1 BETA, Q, LDQ, Z, LDZ, M, PR, PL, DIF, WORK, LWORK,
2 IWORK, LIWORK, INFO)

INTEGER IJOB, N, LDA, LDB, LDQ, LDZ, M, LWORK, IWORK(*),
1 LIWORK, INFO
double precision PR, PL, DIF(*)
complex*16 A(LDA,*), B(LDB,*), ALPHA(*), BETA(*), Q(LDQ,*),
1 Z(LDZ,*), WORK(*)
LOGICAL WANTQ, WANTZ, SELECT(*)

```

The routine may be called by its LAPACK name *ztgsen*.

### 3 Description

F08YUF (ZTGSEN) factorizes the generalized complex  $n$  by  $n$  matrix pair  $(S, T)$  in generalized Schur form, using a unitary equivalence transformation as

$$S = \hat{Q}\hat{S}\hat{Z}^H, \quad T = \hat{Q}\hat{T}\hat{Z}^H,$$

where  $(\hat{S}, \hat{T})$  are also in generalized Schur form and have the selected eigenvalues as the leading diagonal elements. The leading columns of  $Q$  and  $Z$  are the generalized Schur vectors corresponding to the selected eigenvalues and form orthonormal subspaces for the left and right eigenspaces (deflating subspaces) of the pair  $(S, T)$ .

The pair  $(S, T)$  are in generalized Schur form if  $S$  and  $T$  are upper triangular as returned, for example, by F08XNF (ZGGES), or F08XSF (ZHGEQZ) with  $JOB = 'S'$ . The diagonal elements define the generalized eigenvalues  $(\alpha_i, \beta_i)$ , for  $i = 1, 2, \dots, n$  of the pair  $(S, T)$ . The eigenvalues are given by

$$\lambda_i = \alpha_i / \beta_i,$$

but are returned as the pair  $(\alpha_i, \beta_i)$  in order to avoid possible overflow in computing  $\lambda_i$ . Optionally, the routine returns reciprocals of condition number estimates for the selected eigenvalue cluster,  $p$  and  $q$ , the right and left projection norms, and of deflating subspaces,  $Dif_u$  and  $Dif_l$ . For more information see Anderson *et al.* (1999) (Sections 2.4.8 and 4.11).

If  $S$  and  $T$  are the result of a generalized Schur factorization of a matrix pair  $(A, B)$

$$A = QSZ^H, \quad B = QTZ^H$$

then, optionally, the matrices  $Q$  and  $Z$  can be updated as  $Q\hat{Q}$  and  $Z\hat{Z}$ . Note that the condition numbers of the pair  $(S, T)$  are the same as those of the pair  $(A, B)$ .

## 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

## 5 Parameters

- 1: IJOB – INTEGER *Input*  
*On entry:* specifies whether condition numbers are required for the cluster of eigenvalues ( $p$  and  $q$ ) or the deflating subspaces ( $\text{Dif}_u$  and  $\text{Dif}_l$ ).  
 IJOB = 0  
     Only reorder with respect to SELECT. No extras.  
 IJOB = 1  
     Reciprocal of norms of ‘projections’ onto left and right eigenspaces with respect to the selected cluster ( $p$  and  $q$ ).  
 IJOB = 2  
     The upper bounds on  $\text{Dif}_u$  and  $\text{Dif}_l$ .  $F$ -norm-based estimate (DIF(1 : 2)).  
 IJOB = 3  
     Estimate of  $\text{Dif}_u$  and  $\text{Dif}_l$ . 1-norm-based estimate (DIF(1 : 2)). About 5 times as expensive as IJOB = 2.  
 IJOB = 4  
     Compute PL, PR and DIF as in IJOB = 0, 1 and 2. Economic version to get it all.  
 IJOB = 5  
     Compute PL, PR and DIF as in IJOB = 0, 1 and 3.
- 2: WANTQ – LOGICAL *Input*  
*On entry:* if WANTQ = .TRUE., update the left transformation matrix  $Q$ .  
 If WANTQ = .FALSE., do not update  $Q$ .
- 3: WANTZ – LOGICAL *Input*  
*On entry:* if WANTZ = .TRUE., update the right transformation matrix  $Z$ .  
 If WANTZ = .FALSE., do not update  $Z$ .
- 4: SELECT(\*) – LOGICAL array *Input*  
**Note:** the dimension of the array SELECT must be at least  $\max(1, N)$ .  
*On entry:* specifies the eigenvalues in the selected cluster. To select an eigenvalue  $\lambda_j$ , SELECT( $j$ ) must be set to .TRUE..
- 5: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrices  $S$  and  $T$ .  
*Constraint:*  $N \geq 0$ .
- 6: A(LDA,\*) – **complex\*16** array *Input/Output*  
**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .  
*On entry:* the upper triangular matrix  $S$ , in generalized Schur form.

- On exit:* the reordered matrix  $\hat{S}$  in generalized Schur form.
- 7: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F08YUF (ZTGSEN) is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 8: B(LDB,\*) – **complex\*16** array *Input/Output*  
**Note:** the second dimension of the array B must be at least  $\max(1, N)$ .  
*On entry:* the upper triangular matrix  $T$ , in generalized Schur form.  
*On exit:* the reordered matrix  $\hat{T}$  in generalized Schur canonical form.
- 9: LDB – INTEGER *Input*  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F08YUF (ZTGSEN) is called.  
*Constraint:*  $LDB \geq \max(1, N)$ .
- 10: ALPHA(\*) – **complex\*16** array *Output*  
**Note:** the dimension of the array ALPHA must be at least  $\max(1, N)$ .  
*On exit:* see the description of BETA.
- 11: BETA(\*) – **complex\*16** array *Output*  
**Note:** the dimension of the array BETA must be at least  $\max(1, N)$ .  
*On exit:* ALPHA and BETA contain diagonal elements of  $\hat{S}$  and  $\hat{T}$ , respectively, when the pair  $(S, T)$  has been reduced to generalized Schur form.  $ALPHA(i)/BETA(i)$ , for  $i = 1, \dots, N$ , are the eigenvalues.
- 12: Q(LDQ,\*) – **complex\*16** array *Input/Output*  
**Note:** the second dimension of the array Q must be at least  $\max(1, N)$ .  
*On entry:* if WANTQ = .TRUE., the  $n$  by  $n$  matrix  $Q$ .  
*On exit:* if WANTQ = .TRUE., the updated matrix  $Q\hat{Q}$ .  
 If WANTQ = .FALSE., Q is not referenced.
- 13: LDQ – INTEGER *Input*  
*On entry:* the first dimension of the array Q as declared in the (sub)program from which F08YUF (ZTGSEN) is called.  
*Constraints:*  
     if WANTQ = .TRUE.,  $LDQ \geq \max(1, N)$ ;  
      $LDQ \geq 1$  otherwise.
- 14: Z(LDZ,\*) – **complex\*16** array *Input/Output*  
**Note:** the second dimension of the array Z must be at least  $\max(1, N)$ .  
*On entry:* if WANTZ = .TRUE., the  $n$  by  $n$  matrix  $Z$ .  
*On exit:* if WANTZ = .TRUE., the updated matrix  $Z\hat{Z}$ .  
 If WANTZ = .FALSE., Z is not referenced.

- 15: LDZ – INTEGER *Input*  
*On entry:* the first dimension of the array Z as declared in the (sub)program from which F08YUF (ZTGSEN) is called.  
*Constraints:*  
 if WANTZ = .TRUE.,  $LDZ \geq \max(1, N)$ ;  
 $LDZ \geq 1$  otherwise.
- 16: M – INTEGER *Output*  
*On exit:* the dimension of the specified pair of left and right eigenspaces (deflating subspaces) with  $0 \leq M \leq N$ .
- 17: PR – *double precision* *Output*  
 18: PL – *double precision* *Output*  
*On exit:* if IJOB = 1, 4 or 5, PL and PR are lower bounds on the reciprocal of the norm of ‘projections’  $p$  and  $q$  onto left and right eigenspace with respect to the selected cluster.  $0 < PL, PR \leq 1$ .  
 If  $M = 0$  or  $M = N$ ,  $PL = PR = 1$ .  
 If IJOB = 0, 2 or 3, PL and PR are not referenced.
- 19: DIF(\*) – *double precision* array *Output*  
**Note:** the dimension of the array DIF must be at least 2.  
*On exit:* if IJOB  $\geq 2$ , DIF(1 : 2) store the estimates of  $Dif_u$  and  $Dif_l$ .  
 If IJOB = 2 or 4, DIF(1 : 2) are  $F$ -norm-based upper bounds on  $Dif_u$  and  $Dif_l$ .  
 If IJOB = 3 or 5, DIF(1 : 2) are 1-norm-based estimates of  $Dif_u$  and  $Dif_l$ .  
 If  $M = '0'$  or  $'N'$ ,  $DIF(1 : 2) = \|(A, B)\|_F$ .  
 If IJOB = 0 or 1, DIF is not referenced.
- 20: WORK(\*) – *complex\*16* array *Workspace*  
**Note:** the dimension of the array WORK must be at least  $\max(1, LWORK)$ .  
*On exit:* if INFO = 0, WORK(1) returns the minimum LWORK.
- 21: LWORK – INTEGER *Input*  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F08YUF (ZTGSEN) is called.  
 If LWORK = -1, a workspace query is assumed; the routine only calculates the minimum sizes of the WORK and IWORK arrays, returns these values as the first entries of the WORK and IWORK arrays, and no error message related to LWORK or LIWORK is issued.  
*Constraints:*  
 if LWORK  $\neq -1$ ,  
 if IJOB = 1, 2 or 4,  $LWORK \geq \max(1, 2 \times M \times (N - M))$ ;  
 if IJOB = 3 or 5,  $LWORK \geq \max(1, 4 \times M \times (N - M))$ ;  
 $LWORK \geq 1$  otherwise.
- 22: IWORK(\*) – INTEGER array *Workspace*  
**Note:** the dimension of the array IWORK must be at least  $\max(1, LIWORK)$ .  
*On exit:* if INFO = 0, IWORK(1) returns the minimum LIWORK.

23: LIWORK – INTEGER

*Input*

*On entry:* the dimension of the array IWORK as declared in the (sub)program from which F08YUF (ZTGSEN) is called.

If LIWORK = -1, a workspace query is assumed; the routine only calculates the minimum sizes of the WORK and IWORK arrays, returns these values as the first entries of the WORK and IWORK arrays, and no error message related to LWORK or LIWORK is issued.

*Constraints:*

if LIWORK  $\neq$  -1,  
 if IJOB = 1, 2 or 4, LIWORK  $\geq$  N + 2;  
 if IJOB = 3 or 5, LIWORK  $\geq$  max(N + 2, 2  $\times$  M  $\times$  (N - M));  
 LIWORK  $\geq$  1 otherwise.

24: INFO – INTEGER

*Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = -*i*, the *i*th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO = 1

Reordering of (*S*, *T*) failed because the transformed matrix pair ( $\hat{S}, \hat{T}$ ) would be too far from generalized Schur form; the problem is very ill-conditioned. (*S*, *T*) may have been partially reordered. If requested, 0 is returned in DIF(1 : 2), PL and PR.

## 7 Accuracy

The computed generalized Schur form is nearly the exact generalized Schur form for nearby matrices (*S* + *E*) and (*T* + *F*), where

$$\|E\|_2 = O\epsilon\|S\|_2 \quad \text{and} \quad \|F\|_2 = O\epsilon\|T\|_2,$$

and  $\epsilon$  is the *machine precision*. See Anderson *et al.* (1999) (Section 4.11) for further details of error bounds for the generalized nonsymmetric eigenproblem, and for information on the condition numbers returned.

## 8 Further Comments

The real analogue of this routine is F08YGF (DTGSEN).

## 9 Example

To reorder the generalized Schur factors *S* and *T* and update the matrices *Q* and *Z* given by

$$S = \begin{pmatrix} 4.0 + 4.0i & 1.0 + 1.0i & 1.0 + 1.0i & 2.0 - 1.0i \\ 0 & 2.0 + 1.0i & 1.0 + 1.0i & 1.0 + 1.0i \\ 0 & 0 & 2.0 - 1.0i & 1.0 + 1.0i \\ 0 & 0 & 0 & 6.0 - 2.0i \end{pmatrix},$$

$$T = \begin{pmatrix} 2.0 & 1.0 + 1.0i & 1.0 + 1.0i & 3.0 - 1.0i \\ 0 & 1.0 & 2.0 + 1.0i & 1.0 + 1.0i \\ 0 & 0 & 1.0 & 1.0 + 1.0i \\ 0 & 0 & 0 & 2.0 \end{pmatrix},$$

$$Q = \begin{pmatrix} 1.0 & 0 & 0 & 0 \\ 0 & 1.0 & 0 & 0 \\ 0 & 0 & 1.0 & 0 \\ 0 & 0 & 0 & 1.0 \end{pmatrix} \quad \text{and} \quad Z = \begin{pmatrix} 1.0 & 0 & 0 & 0 \\ 0 & 1.0 & 0 & 0 \\ 0 & 0 & 1.0 & 0 \\ 0 & 0 & 0 & 1.0 \end{pmatrix},$$

selecting the second and third generalized eigenvalues to be moved to the leading positions. Bases for the left and right deflating subspaces, and estimates of the condition numbers for the eigenvalues and Frobenius norm based bounds on the condition numbers for the deflating subspaces are also output.

## 9.1 Program Text

```

*      F08YUF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX
PARAMETER       (NMAX=8)
INTEGER          LDQ, LDS, LDT, LDZ, LIWORK, LWORK
PARAMETER       (LDQ=NMAX,LDS=NMAX,LDT=NMAX,LDZ=NMAX,
+              LIWORK=(NMAX*NMAX)/2+2,LWORK=NMAX*NMAX)
*      .. Local Scalars ..
DOUBLE PRECISION PL, PR
INTEGER          I, IFAIL, IJOB, INFO, J, M, N
LOGICAL         WANTQ, WANTZ
*      .. Local Arrays ..
COMPLEX *16     ALPHA(NMAX), BETA(NMAX), Q(LDQ,NMAX),
+              S(LDS,NMAX), T(LDT,NMAX), WORK(LWORK),
+              Z(LDZ,NMAX)
DOUBLE PRECISION DIF(2)
INTEGER          IWORK(LIWORK)
LOGICAL         SELECT(NMAX)
CHARACTER       CLABS(1), RLABS(1)
*      .. External Subroutines ..
EXTERNAL        X04DBF, ZTGSEN
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08YUF Example Program Results'
WRITE (NOUT,*)
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*          Read S, T, Q, Z and the logical array SELECT from data file
*
      READ (NIN,*) ((S(I,J),J=1,N),I=1,N)
      READ (NIN,*) ((T(I,J),J=1,N),I=1,N)
      READ (NIN,*) ((Q(I,J),J=1,N),I=1,N)
      READ (NIN,*) ((Z(I,J),J=1,N),I=1,N)
*
      READ (NIN,*) (SELECT(I),I=1,N)
*
*      Set IJOB, WANTQ and WANTZ
      IJOB = 4
      WANTQ = .TRUE.
      WANTZ = .TRUE.
*
*      Reorder the Schur factors S and T and update the matrices
*      Q and Z
*

```

```

      CALL ZTGSEN(IJOB,WANTQ,WANTZ,SELECT,N,S,LDS,T,LDT,ALPHA,BETA,Q,
+             LDQ,Z,LDZ,M,PL,PR,DIF,WORK,LWORK,IWORK,LIWORK,INFO)
      IF (INFO.NE.0) THEN
        WRITE (NOUT,99999) INFO
        WRITE (NOUT,*)
      END IF

*
*   Print reordered generalized Schur form
*
      IFAIL = 0
      CALL X04DBF('General',' ',N,N,S,LDS,'Bracketed','F7.4',
+             'Reordered Schur matrix S','Integer',RLABS,
+             'Integer',CLABS,80,0,IFAIL)
*
      WRITE (NOUT,*)
      IFAIL = 0
      CALL X04DBF('General',' ',N,N,T,LDT,'Bracketed','F7.4',
+             'Reordered Schur matrix T','Integer',RLABS,
+             'Integer',CLABS,80,0,IFAIL)
*
*   Print deflating subspaces
*
      WRITE (NOUT,*)
      IFAIL = 0
      CALL X04DBF('General',' ',N,M,Q,LDQ,'Bracketed','F7.4',
+             'Basis of left deflating invariant subspace',
+             'Integer',RLABS,'Integer',CLABS,80,0,IFAIL)
      WRITE (NOUT,*)
      IFAIL = 0
      CALL X04DBF('General',' ',N,M,Z,LDZ,'Bracketed','F7.4',
+             'Basis of right deflating invariant subspace',
+             'Integer',RLABS,'Integer',CLABS,80,0,IFAIL)
*
*   Print norm estimates and F-norm upper bounds
*
      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'Norm estimate of projection onto',
+             ' left eigenspace for selected cluster', 1.0D0/PL
      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'Norm estimate of projection onto',
+             ' right eigenspace for selected cluster', 1.0D0/PR
      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'F-norm based upper bound on', ' Difu',
+             DIF(1)
      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'F-norm based upper bound on', ' Difl',
+             DIF(2)
      ELSE
        WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP

*
99999 FORMAT (' Reordering could not be completed. INFO = ',I3)
99998 FORMAT (1X,2A,/1X,1P,E10.2)
      END

```

## 9.2 Program Data

F08YUF Example Program Data

```

4
( 4.0, 4.0) ( 1.0, 1.0) ( 1.0, 1.0) ( 2.0,-1.0)
( 0.0, 0.0) ( 2.0, 1.0) ( 1.0, 1.0) ( 1.0, 1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 2.0,-1.0) ( 1.0, 1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0) ( 6.0,-2.0)
( 2.0, 0.0) ( 1.0, 1.0) ( 1.0, 1.0) ( 3.0,-1.0)
( 0.0, 0.0) ( 1.0, 0.0) ( 2.0, 1.0) ( 1.0, 1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 1.0, 0.0) ( 1.0, 1.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0) ( 2.0, 0.0)
( 1.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0)
( 0.0, 0.0) ( 1.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 1.0, 0.0) ( 0.0, 0.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0) ( 1.0, 0.0)
( 1.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0)
( 0.0, 0.0) ( 1.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 1.0, 0.0) ( 0.0, 0.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0) ( 1.0, 0.0)
( 0.0, 0.0) ( 0.0, 0.0) ( 0.0, 0.0) ( 1.0, 0.0)
F           T           T           F
:Value of N
:End of matrix S
:End of matrix T
:End of matrix Q
:End of matrix Z
:End of SELECT

```

## 9.3 Program Results

F08YUF Example Program Results

Reordered Schur matrix S

```

1           2           3           4
1 ( 4.6904, 2.3452) (-2.1563, 0.1192) ( 1.9599,-0.5174) ( 1.8091,-1.2060)
2 ( 0.0000, 0.0000) ( 2.0084,-1.0042) ( 0.9161,-0.2762) ( 1.8574,-0.5326)
3 ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 1.6985, 1.6985) ( 0.1270, 0.7231)
4 ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 6.0000,-2.0000)

```

Reordered Schur matrix T

```

1           2           3           4
1 ( 2.3452, 0.0000) (-0.6181, 1.8237) ( 0.9290, 0.5409) ( 2.7136,-1.5076)
2 ( 0.0000, 0.0000) ( 1.0042, 0.0000) ( 1.2251,-1.1857) ( 1.8541,-0.2929)
3 ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 0.8492, 0.0000) ( 0.1435, 0.9053)
4 ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 2.0000, 0.0000)

```

Basis of left deflating invariant subspace

```

1           2
1 ( 0.9045, 0.3015) (-0.0033,-0.2397)
2 ( 0.3015, 0.0000) ( 0.2497, 0.7157)
3 ( 0.0000, 0.0000) ( 0.0549, 0.6042)
4 ( 0.0000, 0.0000) ( 0.0000, 0.0000)

```

Basis of right deflating invariant subspace

```

1           2
1 ( 0.7071, 0.0000) (-0.5607, 0.0000)
2 ( 0.7071, 0.0000) ( 0.5607, 0.0000)
3 ( 0.0000, 0.0000) ( 0.0552, 0.6067)
4 ( 0.0000, 0.0000) ( 0.0000, 0.0000)

```

Norm estimate of projection onto left eigenspace for selected cluster  
8.90E+00

Norm estimate of projection onto right eigenspace for selected cluster  
7.02E+00

F-norm based upper bound on Difu  
2.18E-01

F-norm based upper bound on Difl  
2.62E-01